

第3章 物理学編

物理の授業で学ぶ内容は、様々な法則や定理が数式で表されることが多いのですが、その意味するところをイメージとして具体的に描くことは容易なことではありません。特に、力学の中心にある運動方程式は、はじめて物理を学ぶ者にとって容易に理解できない部分です。ここでは力学の内容に限った以下のテーマについて、十進 BASIC による数値解析を実行することにより、イメージとして捉えにくい内容をシミュレーションで紹介しましょう。

1. 放物運動
2. 単振動の運動
3. 減衰振動
4. 人工衛星の軌道
5. 制限3体問題
6. カオスをつくる

上記テーマ1～4は低学年でも理解できる内容です。5. 制限3体問題と6. カオスは少しレベルが高く、授業で学ぶ機会も少ないかも知れませんが、興味深い問題ですから挑戦するつもりで読んで下さい。末尾の文献を参考に自習することをお勧めます。各節に掲げた十進 BASIC プログラム例を実行するだけでもそのおもしろさ分かるでしょう。十進 BASIC の長所は、何よりもそのプログラム言語の知識を十分習得していなくても、ヘルプを頼りに目的の数値処理ができるところにあります。この例題を通して、プログラム言語の長所である簡便さを分かっていたいただければ、今後の

研究, 教育に必ず役に立つことでしょう.

3.1 放物運動

3.1.1 空気抵抗がない場合

質量 m の物体を水平と角 θ をなす方向に初速度 v_0 で投げ上げたとき, 物体はどのような軌道を描くでしょうか [1]. 物体の運動方程式は

$$m a_x = 0, \quad m a_y = -mg \quad (3.1)$$

で与えられます [3]. この運動方程式を初期条件, $t = 0 : x = 0, y = 0, v_x = v_0 \cos \theta, v_y = v_0 \sin \theta$ で解くと, 速度の成分は

$$v_x = v_0 \cos \theta, \quad v_y = v_0 \sin \theta t - gt \quad (3.2)$$

で与えられ, 位置は以下ようになります.

$$x = v_0 t \cos \theta, \quad y = v_0 t \sin \theta - \frac{1}{2} g t^2 \quad (3.3)$$

最後の2式より t を消去すると放物線軌道

$$y = x \tan \theta - \frac{g}{2v_0^2 \cos^2 \theta} x^2 \quad (3.4)$$

が導かれます. 初速度を $v_0 = 5.0[\text{m/s}]$ として, 水平線と角度 $\theta = 15^\circ, 30^\circ, 45^\circ, 60^\circ, 75^\circ$ で投げ出した場合の軌道を描くプログラムとその出力結果を図 3.1 に示します.

プログラム例

```
100 ! 1.1 空気抵抗がない場合の放物運動.
110 ! 初速度 VO[m/s] で水平と角度  °の方向に投げ上げる
120 OPTION ANGLE DEGREES
130 SET WINDOW -1, 4, -0.2, 1.5
140 DRAW GRID(0.5, 0.2)
```




図 3.1: 空気抵抗のない場合の放物運動

```
150 LET g=9.8
160 LET v0=5.0
170 FOR theta=15 to 75 step 15
180   FOR x=0 TO 20 STEP 0.001
190     LET Y=X*TAN(THETA)-0.5*g*(x/(v0*COS(theta)))^2
200     IF Y>=0 THEN
210       PLOT LINES: x, Y;
220     ELSE
225       PLOT LINES
230       GOTO 260
240     END IF
250   NEXT X
260 NEXT theta
270 END
```

問題 1 水平となす角度 θ をいろいろ変えて、水平到達距離が最大になる角度を求めてみよう。

3.1.2 モンキーと椰子の実

地上からある高さの位置の木の上に猿を見つけた．その猿を目掛けて椰子の実を投げたところ，猿は投げると同時に木から落ちた．しかし，椰子の実はみごと猿に命中して，ご馳走にありつけた．この状況は，実は空気の抵抗を無視して，猿が自由落下すると仮定すれば猿の位置によらず，また，椰子の実を投げる速さによらず命中します．これを証明するのは容易なことです [1] が，ここでは椰子の実と猿を質点と考えて（空気抵抗を無視しているので質点でも同じ），この状況をシミュレーションでたどってみます．椰子の実の運動は放物運動であり，猿の運動は自由落下運動です．そのプログラム例と出力結果を図 3.2 に示します．

プログラム例

```

100 ! 1.2 モンキーと椰子の実
110 ! 猿が目掛けて（水平と角  °）初速 V0[m/s] で椰子の実を投
120 ! げる．同時に，水平距離 a [m] 離れた高さ a tan  °の木か
130 ! ら猿が落下する．見事，椰子の実は猿に命中する．
140 OPTION ANGLE DEGREES
150 SET WINDOW -1,6,-3,9
160 SET POINT STYLE 4
170 DRAW GRID(1.,1.)
180 LET g=9.8
190 LET theta=50
200 ! --- 水平距離 a と初速度 V0 を変えて実験してみる．---
210 FOR I=1 TO 3
220     LET a=2.0+I
230     LET V0=3.0+2.0*I
240     LET t=0
250     PLOT POINTS: a, a*TAN(theta)
260     DO
265 ! ----- 椰子の実の運動 -----
270         LET X=V0*COS(theta)*t
280         LET Y=V0*SIN(theta)*t-0.5*g*t*t
285 ! ----- 猿の落下運動 -----

```




図 3.2: モンキーと椰子の実

```

290     LET XX=a
300     LET YY=a*TAN(theta)-0.5*g*t*t
310     PLOT LINES: X, Y
320     PLOT LINES: XX, YY
330     LET t=t+0.00001
340     LOOP WHILE X < a
350     PLOT POINTS: X, Y
360     PRINT "猿が"; a; "m はなれた, 水平と"; theta; " °の方向
の位置から落下するとき, 落下"
370     PRINT "と同時に椰子の実を初速"; V0; " m/s で投げると"; t; "
秒後に必ず命中する ."
380     PRINT
390 NEXT I
400 END

```

プログラム例では時間刻みを 10^{-5} にとっていますが、小さくするに従って運動の変化をスローモーションで再現することができます。また、図 3.2 から椰子の実の投げる位置を地上の点とすると、距離 $a = 3[\text{m}]$ のとき、初速度 v_0 が $5[\text{m/s}]$ では地上で命中しないことが分かります。

問題 2 猿までの水平距離を固定し、椰子の実の初速 v_0 をいろいろ変

えて、命中するかどうか調べてみよう。

3.1.3 速さに比例する抵抗が働く場合

放物運動で速さに比例する空気などによる抵抗（粘性抵抗）がある場合に、運動が抵抗がない場合に比べ、どのようにずれてくるのかシミュレーションでたどってみよう。 v_0, θ は抵抗のない場合と同様に定義し、抵抗力を $c\vec{v}$ で表すと、物体の運動方程式は

$$m a_x = -c v_x, \quad m a_y = -m g - c v_y \quad (3.5)$$

のように表されます。 $a_x = dv_x/dt$, $a_y = dv_y/dt$ ですから、上の微分方程式は簡単に積分できて、最終的に

$$\begin{cases} x = v_0 \cos \theta (1 - e^{-kt}) / k \\ y = -gt/k + (v_0 \sin \theta + g/k)(1 - e^{-kt}) / k \end{cases} \quad (3.6)$$

のように求めることができます [4]。ここで、 $k = c/m$ 。時間 t を変化させて、地上に落下するまでの軌道を求めるプログラム例を以下に示します。

```
100 ! 1.3 放物運動2 (速さに比例する抵抗がある場合)
110 ! 10進BASIC プログラム例と出力結果
120 SET WINDOW -0.2, 1.0, -0.2, 0.5
130 LET V0=3.0
140 LET TH=45
150 LET VS=V0*SIN(PI*TH/180)
160 LET VC=V0*COS(PI*TH/180)
170 DRAW AXES(0.1, 0.05)
180 PRINT "K: (空気抵抗)c/m, X: 水平到達距離, T: 所要時間"
190 PRINT
200 FOR KK=1 TO 10 STEP 1
210   FOR T=0.0 TO 10.0 STEP 0.001
220     LET P=(1-EXP(-KK*T))/KK
230     LET X=VC*P
240     LET Y=-9.8*T/KK+P*(VS+9.8/KK)
```




図 3.3: 抵抗のある場合の放物運動

```

250     PLOT X, Y;
260     IF Y<0 THEN
270 PRINT USING "K=-%.#  X=  -%.####  Y=  -%.#####  T=  -%.####":
      KK, X, Y, T
280         EXIT FOR
290     END IF
300 NEXT T
310 PLOT LINES
320 NEXT KK
330 END

```

物体の飛行の軌跡は、抵抗の大きさによって図 3.3 のように変化します。

問題 3 水平となす角度 θ を固定して、水平到達距離 x 、および所要時間 T が抵抗 $k(=c/m)$ の関数としてどのように変化するか調べてみよう。

3.2 単振動の運動

3.2.1 位置の時間変化

ばねの一端を固定して、他端に質量 m の質点を結びつけて、少し引っ張ってから手放すと振動運動を始めます。このとき、質点にはフックの法則に従う復元力 $f = -kx$ が働きます。 k はばねの強さを示すばね定数です。このような単純な振動運動を単振動とよびます [2]。その運動方程式は

$$m \frac{d^2x}{dt^2} = -kx \quad (3.7)$$

で表されます [3]。この2階の微分方程式の解は

$$x = A \sin(\omega t + \alpha), \quad \omega = \sqrt{k/m} \quad (3.8)$$

で与えられます。この正弦関数を描くことは簡単なことですが、今後の数値実験のために、微分方程式そのものを数値的に解くことをしてみます。ここではオイラー法とよばれる最も簡単な方法で解いてみます。

時間刻み Δt を十分小さくにとって、その間に速度が v から $v + \Delta v$ に、質点の位置が x から $x + \Delta x$ に変化したとします。 $\Delta v = a \Delta t$ ($a = d^2x/dt^2 = -k/m$) ですから v, x の初期値を与えればこれらの値の時間変化を求めることができます。ばね定数 k を変えて、質点の位置 x の時間変化を与えるプログラム例を以下に示します。ここでは簡単のために質量を $m = 1$ 、初期値を $x = a, v = 0$ にとりました。

100 ! 2.1 単振動の運動 (位置の時間変化)

110 ! 運動方程式を数値的に解く。ばね定数 K の効果

115 ! K: 0.5(黒), 1.0(青), 1.5(緑), 2.0(赤)

120 SET WINDOW -0.5, 1.1, -1.5, 1.5

130 DRAW grid(1, 0.2)

140 ! 質量 m=1 とする。

150 ! 初期値 X=A; V=0

160 ! K : ばね定数

170 FOR IK=1 TO 4

! 4 種類の強さのばね定数




図 3.4: 単振動：位置の時間変化

```

180 LET K=IK*0.5
190 LET A=1
200 LET X=A ! 位置 x の初期値
210 LET V=0 ! 速度の初期値
220 LET t=0
230 SET COLOR IK ! ばねの強さによって描画を色で識別
240 LET dt= 0.01 ! 時間刻み
250 LET DV=-K*X*dt
260 LET V=V+DV
270 LET X=X+V*dt
280 LET t=t+dt
290 PLOT t,X
300 IF t>10 THEN GOTO 320
310 GOTO 250
320 NEXT IK
330 END

```

図 3.4 から単振動の運動が正確に正弦関数で表されることが分かります。
 そして、このような簡単な手法で、微分方程式の数値解が得られるのです。

問題 4 単振動の周期は $T = 2\pi \sqrt{m/k}$ で与えられます [2] . このことを上記プログラムのばね定数 k をさらに細かく変化させることによって確かめてみよう .

3.2.2 ばねのエネルギー

単振動は , ばね (ばね定数 k) に結わえ付けられた質量 m の質点の運動でした . ここでは , 単振動を行う質点の運動エネルギー $K = 1/2 m v^2$, ばねの位置エネルギー $U = 1/2 k x^2$ とその和である力学的エネルギー $E = K + U$ を時間の関数としてその変化を考えてみます . 縦軸のスケールは異なりますが , これに対応した質点の位置 x を参考のために表すことにして , そのプログラム例を以下に示します .

```

100 ! 2.2 ばねのエネルギー
110 ! 運動方程式から数値的に求める
120 ! 運動エネルギー K と位置エネルギー U の関係
130 ! K: 青, U: 緑, 力学的エネルギー E=K+U: 赤, |X|: 黒
140 SET WINDOW -0.5, 11, -.05, 1.05
150 DRAW grid(1, 0.1)
160 ! 質量 m=1 , ばね定数 CK=1 とする
170 ! 位置と速度の初期値 X=A; V=0
180 LET CK=1.0
190 LET A=1
200 LET X=A
210 LET V=0
220 LET t=0
230 LET dt= 0.001
240 LET DV=-CK*X*dt
250 LET V=V+DV
260 LET X=X+V*dt
270 LET t=t+dt
280 SET COLOR 1
290 PLOT t, ABS(X) ! 変位の大きさ

```




図 3.5: 質点の位置, 運動エネルギー, 位置エネルギー, 力学的エネルギーの時間変化

```

300 LET K=V*V/2           ! 運動エネルギー
310 SET COLOR 2
320 PLOT t,K
330 LET U=CK*X*X/2       ! 位置エネルギー
340 SET COLOR 3
350 PLOT t,U
360 LET E=K+U           ! 力学的エネルギー
370 SET COLOR 4
380 PLOT t,E
390 IF t>10 THEN STOP
400 GOTO 250
410 END

```

問題 5 質点の位置と運動エネルギー, 位置エネルギーがどんな関係にあるか, 図 3.5 を見ながら考えてみよう. プログラムの時間刻みを変えて計算させて, 計算精度とはどんな関係があるだろうか.

3.3 減衰振動

単振動の運動は、質量 m の質点をばねが自然な長さにある状態から x だけ変位させたとき、これに比例する復元力が働く（フックの法則）ことよって起こる振動運動でしたが、空気などまわりの媒質からこの運動を妨げる粘性抵抗が働くとき、その運動は単振動の運動からどのように変化するだろうか [4]。直感的に単振動が時間と共に減衰していくことが予測されます。このことをシミュレーションで確認してみよう。まず、この粘性抵抗が速度 v に比例するとして、この場合の質点のしたがう運動方程式は

$$m \frac{d^2 x}{dt^2} = -kx - cv \quad (3.9)$$

で与えられます。2 節の場合と同様に、質量 $m = 1$ にとり、同じ初期値を用います。さらに、ばね定数を $k = 1$ にして、粘性抵抗の影響をその係数 c を変えることによって調べてみよう。微分方程式は、以前の例と同様にオイラー法を用います。プログラム例を以下に示します。

```

100 ! 3.1 減衰振動
110 ! 運動方程式を数値的に解く。粘性抵抗 C の効果
115 ! C: 0.2(黒), 0.4(青), 0.6(緑), 0.8(赤)
120 SET WINDOW -0.5, 20, -1.5, 1.5
130 DRAW grid(1, 0.2)
140 ! 質量 m=1 とする
150 ! 初期値 X=A; V=0
160 ! K : ばね定数, C : 粘性抵抗
165 LET K=1
166 LET A=1
170 FOR IK=1 TO 4
180 LET C=IK*0.2
200 LET X=A
210 LET V=0
220 LET t=0
230 SET COLOR IK
240 LET dt= 0.01
250 LET DV=- (K*X+C*V)*dt

```




図 3.6: 減衰振動：抵抗の大きさと減衰の変化

```

260 LET V=V+DV
270 LET X=X+V*dt
280 LET t=t+dt
290 PLOT t,X
300 IF t>20 THEN GOTO 320
310 GOTO 250
320 NEXT IK
330 END

```

図 3.6 に示した例は，粘性抵抗が比較的小さい場合です．4 種類の減衰振動は，粘性係数 c が大きくなるにしたがい減衰の度合いが大きくなっています．さらに大きくなる ($c > 2\sqrt{mk}$) と振動しなくてそのまま変位 x は平衡点の 0 に近づいていきます．この状態は過減衰とよばれます．

問題 6 振動が減衰振動から臨界減衰 ($c = 2\sqrt{mk}$) を経て，過減衰とよばれる状態に変化する状況を，粘性抵抗係数 c を変化させることによ

て調べてみよう。

問題 7 減衰振動を行う場合 ($c < 2\sqrt{mk}$) , ばね定数 k , 抵抗係数 c を固定して , 前節で調べたばねのエネルギーの時間変化を求め , 前節 $c = 0$ の場合と比べてみよう。

3.4 人工衛星の軌道

地球から打ち上げた人工衛星がどのような軌道を描くか求めてみよう。質量 m の人工衛星は地球の万有引力によって , その中心方向に引っ張られます。他方 , 回転運動によって , 外向きに遠心力が作用します。人工衛星の従う運動方程式は以下のように書くことができます [4]。

$$m \frac{d\vec{v}}{dt} = - \frac{G m M}{r^2} \frac{\vec{r}}{r} \quad (3.10)$$

運動の記述を簡単にするために , 地球の半径 6,370km を 1 の大きさにとります。万有引力定数 G と地球の質量 M の積 $GM = 3.986 \times 10^{14} [\text{m}^3/\text{s}^2]$ を 1 にするために , 上記の長さの定義から時間について , 805 sec を 1 とする定義が得られます。このように , 長さと時間を定義すれば , 上記の運動方程式の x 成分 , y 成分から加速度の成分 (a_x, a_y) はそれぞれ

$$\begin{aligned} a_x &= dv_x/dt = -x/r^3 \\ a_y &= dv_y/dt = -y/r^3 \end{aligned} \quad (3.11)$$

と表されます。従って , 速度の x 成分 , y 成分は微小な時間 Δt をとって , 以下のようなオイラー法による繰り返し計算

$$\begin{aligned} v_x(t + \Delta t) &= v_x(t) + a_x \Delta t \\ v_y(t + \Delta t) &= v_y(t) + a_y \Delta t \end{aligned} \quad (3.12)$$

を実行することにより , 数値的に求めることができます。人工衛星の位置 (x, y) は (3.12) 式の加速度成分を速度成分 (v_x, v_y) で置き換えた類似な

関係

$$\begin{aligned}x(t + \Delta t) &= x(t) + v_x \Delta t \\y(t + \Delta t) &= y(t) + v_y \Delta t\end{aligned}\quad (3.13)$$

から数値的に求めることができます。そのプログラム例と実行結果 (図 3.7) を以下に示します。

```
100 ! 4.1 人工衛星の軌道
110 ! 長さの単位: 地球の半径 R0=6370Km を 1 Length
120 ! GM = 3.986 × 10^14 [m^3/s^2] を 1
130 ! 時間の単位: 805sec を 1 Time
140 ! 時間の刻み: DT
150 SET WINDOW -6, 6, -6, 6
160 DRAW AXES
170 LET R0=6370
180 DRAW DISK
190 PRINT "人工衛星の軌道"
200 ! Initial Value -----
210 INPUT PROMPT "発射点の高度 地上 H [Km]=": H
220 PRINT
230 for i=1 to 3
240   LET V0=7.0+1.0*(i-1)
250   LET DT=0.1
260   LET X=0
270   LET Y=(R0+H)/R0
280   LET Y0=Y
290   LET VX=0.1264*V0
300   LET VY=0
310   LET T=0
320   ! -----
330   DO
340     LET R=SQR(X*X+Y*Y)
350     PLOT LINES: X, Y
360     IF R < 1 THEN
370       GOTO 490
380     ELSE
390       LET AX=-X/R^3
400       LET AY=-Y/R^3
410       LET VX=VX+AX*DT
```

```

人工衛星の軌道
発射点の高度 地上 H [Km]=1000

初速度 V0[Km/s]= 7 のとき, 墜落: 落下時刻[sec]は 1610
初速度 V0[Km/s]= 8 のとき, 周期[sec]は 8452.5
初速度 V0[Km/s]= 9 のとき, 周期[sec]は 17629.5

```




図 3.7: 人工衛星の軌道

```

420      LET  VY=VY+AY*DT
430      LET  X=X+VX*DT
440      LET  Y=Y+VY*DT
450      IF ABS((Y-Y0)/Y0) < 1.0E-3 THEN GOTO 530
460      LET  T=T+DT
470      END IF
480      LOOP
490      PRINT "初速度 V0[Km/s]="; V0;
500      PRINT "のとき, 墜落: 落下時刻 [sec] は"; T*805
510      PRINT
520      GOTO 560
530      PRINT "初速度 V0[Km/s]="; V0;
540      PRINT "のとき, 周期 [sec] は"; T*805
550      PRINT
560 next i
570 END

```

このプログラム例では地上から直接人工衛星を発射するのではなく、地上の高さ H km から水平に v_0 km/s で発射するようにしています。

問題 8 いろいろな楕円軌道のシミュレーションから長半径 a と周期 T の関係を求めて、Kepler の第 3 法則 $a^3 \propto T^2$ が成り立っているか確かめてみよう。

問題 9 地上から水平に対して、角度 θ で人工衛星を打ち上げたときの軌道を描くプログラムをつくり、実行させてみよう。

3.5 制限3体問題

この節ではもう少し複雑な天体運動 [5],[6] について考えてみます。人工衛星の運動が地球の万有引力だけで運動しているとき、その運動は 2 体問題として完全に解くことができます。しかし、人工衛星が月の近くを運動するような場合には地球の引力だけでなく、月の引力も働き運動は簡単ではありません。仮に、人工衛星でなくて、小さな天体であれば月や地球の運動まで影響を受ける可能性があるので、さらに運動は複雑になります。このように、物体の運動は、3 体以上の天体が万有引力によって相互作用していると、これを正確に解くことができません。但し、人工衛星のような小さな天体であれば、これによって地球や月の運動が変更されることはないので、近似的にこの天体の軌道を求めることができます。このような 3 体問題を特に、制限 3 体問題とよびます。

2 つの大きな天体 M_1, M_2 が距離 D 離れて、角速度 ω で xy 平面上を運動しているものとします。この 2 つの天体には万有引力が働いていますから、これにつり合うだけの外向きの力（遠心力）が必要であるため、2 つの天体は鉄アレイのように回転しているのです。2 つの天体の近くに質量 m の小さな第 3 の天体があるとき、この天体はどのような運動をするでしょうか。問題を簡単にするために、天体 M_1, M_2 を x 軸上の点 $P_1 = (x_1, 0)$ および $P_2 = (x_2, 0)$ に固定させて、第 3 の天体 m は xy 平面上の点 $P = (x, y)$




図 3.8: 制限 3 体問題：運動座標系と 3 つの天体

にあるものとします．そして，この平面が z 軸の周りに角速度 ω で回転していると考えればよいのです．このような座標系は回転座標系と呼ばれて，この座標系で運動する質点には，通常の力の他に回転座標系に伴う，遠心力，コリオリ力が作用します．

第 3 の質量 m の天体の運動方程式は

$$m \frac{d^2 \vec{r}}{dt^2} = \vec{F} + 2m \vec{r} \times \vec{\omega} + m \omega^2 \vec{r} \quad (3.14)$$

のように書くことができます [4]．右辺，第 1 項は 2 つの天体からの万有引力の和で，第 2 項はコリオリ力，第 3 項は遠心力を表しています．第 3 の天体の運動が xy 平面に限られることや角速度が $\vec{\omega} = (0, 0, \omega)$ であることを考慮すると，上記の運動方程式は次のような 2 つの連立した微分方程式に書きなおすことができます．

$$\begin{aligned} d^2 x / dt^2 - 2\omega dy / dt - \omega^2 x &= F_x / m = f_x \\ d^2 y / dt^2 + 2\omega dx / dt - \omega^2 y &= F_y / m = f_y \end{aligned} \quad (3.15)$$

但し, f_x, f_y は2つの万有引力による力の成分に関連して

$$f_x = -\frac{GM_1\{x - DM_2/(M_1 + M_2)\}}{r_1^3} - \frac{GM_2\{x + DM_1/(M_1 + M_2)\}}{r_2^3} \quad (3.16)$$

$$f_y = -\frac{GM_1y}{r_1^3} - \frac{GM_2y}{r_2^3} \quad (3.17)$$

で表されます. r_1, r_2 はそれぞれ第3の天体 m とこれらの天体との距離で

$$\begin{aligned} r_1^2 &= \{x - DM_2/(M_1 + M_2)\}^2 + y^2 \\ r_2^2 &= \{x + DM_1/(M_1 + M_2)\}^2 + y^2 \end{aligned} \quad (3.18)$$

で与えられます. f_x, f_y の式は万有引力定数 G を含み, 複雑な形をしていますから, 人工衛星の軌道計算で行ったと類似の変数変換

$$x = DX, y = DY, M_1 = (M_1 + M_2)M'_1, M_2 = (M_1 + M_2)M'_2 \quad (3.19)$$

および

$$t = \left[\frac{D^3}{G(M_1 + M_2)} \right]^{1/2} t' \quad (3.20)$$

を行うと X, Y や M'_1, M'_2 はそれぞれ無次元化された座標や質量で, t' は無次元化された時間になります. ここで, 2つの天体 M_1, M_2 の力のつり合いから得られる関係式

$$G(M_1 + M_2) = D^3 \omega^2 \quad (3.21)$$

を用いると, 連立微分方程式 (3.14) は以下のように簡単になります.

$$\begin{aligned} d^2X/dt'^2 - 2dY/dt' - X &= f_X \\ d^2Y/dt'^2 + 2dX/dt' - Y &= f_Y \end{aligned} \quad (3.22)$$

$$f_X = -\frac{M'_1(X - M_2)}{R_1^3} - \frac{M_2(X + M'_1)}{R_2^3} \quad (3.23)$$

$$f_Y = -\frac{M'_1Y}{R_1^3} - \frac{M'_2Y}{R_2^3} \quad (3.24)$$

$$\begin{aligned} R_1^2 &= (X - M'_2)^2 + Y^2 \\ R_2^2 &= (X + M'_1)^2 + Y^2 \end{aligned} \quad (3.25)$$

2変数の連立した2階微分方程式(3.22)をオイラー法より精度のよい4次のルンゲ・クッタ法で解析して求めます。この解析法で第3の天体の軌道を求めるプログラム例を以下に示します。

```

100 ! 5.1 制限3体問題
110 DIM X(4), Y(4), K(4, 4)
120 DEF FNFX(X, Y)=-M1*(X-M2)/FNR(X-M2, Y)^3-M2*(X+M1)/FNR(X+M1, Y)^3
130 DEF FNFY(X, Y)=-M1*Y/FNR(X-M2, Y)^3-M2*Y/FNR(X+M1, Y)^3
140 DEF FNR(X, Y)=SQR(X*X+Y*Y)
150 DEF FNV(X, Y)=M1/FNR(X-M2, Y)+M2/FNR(X+M1, Y)
160 SET WINDOW -1.3, 1.3, -1.3, 1.3
170 LET M1=0.5 ! 第1天体の質量
180 LET M2=0.5 ! 第2天体の質量
190 LET X(1)=-0.3 ! 第3天体のx座標(初期値)
200 LET X(2)=0 ! 第3天体のy座標(初期値)
210 LET X(3)=0.55 ! 第3天体の速度x成分(初期値)
220 LET C2=3.8 ! エネルギー初期値
230 LET X(4)=SQR(X(1)*X(1)+X(2)*X(2)+2*FNV(X(1), X(2))-C2)
240 LET H0=0.001
250 LET H=H0 ! H: 刻み幅
260 PLOT LINES: -1.1, -0.6; 1.1, -0.6; 1.1, 0.6; -1.1, 0.6; -1.1, -0.6
265 SET COLOR 4
270 DRAW disk WITH SCALE(0.02)*SHIFT(-M1, 0) ! M1の天体位置
280 DRAW disk WITH SCALE(0.02)*SHIFT(M2, 0) ! M2の天体位置
285 SET COLOR 1
290 PLOT X(1), X(2) ! 第3天体の最初の位置
295 ! % ----- ルンゲ・クッタ法 ----- %
300 FOR I=1 TO 4
310 LET Y(I)=X(I)
320 NEXT I
330 LET I=1
340 GOSUB 690
350 FOR I=1 TO 4
360 LET Y(I)=X(I)+H/2*K(I, 1)
370 NEXT I
380 LET I=2
390 GOSUB 690
400 FOR I=1 TO 4

```

```

410 LET Y(1)=X(1)+H/2*K(I,2)
420 NEXT I
430 LET I=3
440 GOSUB 690
450 FOR I=1 TO 4
460 LET Y(1)=X(1)+H*K(I,3)
470 NEXT I
480 LET I=4
490 GOSUB 690
500 FOR I=1 TO 4
510 LET Y(1)=X(1)+H/6*(K(I,1)+2*(K(I,2)+K(I,3))+K(I,4))
520 NEXT I
530 LET R2=Y(1)*Y(1)+Y(2)*Y(2)
540 LET V2=Y(3)*Y(3)+Y(4)*Y(4)
550 LET E=C2+V2-R2-2*FNV(Y(1),Y(2))
560 LET A=SQR(1-E/V2)
570 LET X(1)=Y(1)
580 LET X(2)=Y(2)
590 LET X(3)=A*Y(3)
600 LET X(4)=A*Y(4)
610 IF V2>10 THEN ! 速度が速くなると精度が下がる
620 LET H=H0/SQR(V2) ! ので刻み幅を細かくする
630 ELSE
640 LET H=H0
650 END IF
660 PLOT LINES: X(1),X(2);
670 GOTO 300
680 STOP
690 LET K(1,1)=Y(3) ! k1 の定義
700 LET K(2,1)=Y(4) ! k2 の定義
710 LET K(3,1)=Y(1)+2*Y(4)+FNFX(Y(1),Y(2)) ! k3 の定義
720 LET K(4,1)=Y(1)-2*Y(3)+FNFY(Y(1),Y(2)) ! k4 の定義
730 RETURN
740 END

```

上記のプログラムにおいて、 H は刻みの大きさを計算精度に関係します。 H を小さくすると精度は上がりますが、それだけ計算量が増えて時間がか




図 3.9: 制限3体問題：初期条件によっては第3の天体はカオス軌道になる

かります。しかし、そのことによって第3の天体の運動の様子をより詳しく観察できます。プログラムの実行結果、図 3.9 から第3の天体が定まらないカオス軌道をしていることがわかります。そして、この天体の運動がエネルギー保存則によってある領域に限られていることも分かるでしょう。第2，第3の天体の近くでは万有引力によって速度が増しています。第3の天体が第2，第3の天体の上を通過していますが、ここでは天体の大きさを無視して点として扱っています。プログラムは無限に終了しない形式に書かれています。運動の様子を把握した時点で中断のキイを押して、計算を止めるようにしてください。

問題 10 エネルギー初期値（上記プログラムの $C2$ ）を変えることによって、第3の天体の軌道は様々な振る舞いをします。また、刻みの大きさ H を変えてプログラムを実行してみよう。

3.6 カオスをつくる

カオスの発生は非線形現象が関係しています．簡単な数式を使うことによって人工的にカオスを作り出すことができます．写像は，一般的に

$$x_{n+1} = f(x_n) \quad (3.26)$$

のような差分方程式を実行することで求めることができますが，このような繰り返し計算は，コンピュータが最も得意とするところで，最近のパソコンでは100万回の繰り返し計算でも直ちに結果を与えてくれます．以下に与えるロジスティック写像，対数写像は共に単純な写像関数ですが，多数回の繰り返し計算がどのような結果を与えるのか，全くと云ってよいほど予測が付きません．そこにまた非線形現象の複雑さとおもしろさが潜んでいると云えるでしょう．

3.6.1 ロジスティック写像

オーストラリア生まれの生物学者ロバート・メイは差分方程式を用いて，昆虫の個体数の増殖を調べていましたが，そのモデルに用いた関数がロジスティック写像と呼ばれる差分方程式

$$x_{n+1} = a x_n (1 - x_n) \quad (3.27)$$

です．ロバート・メイは，生徒が電卓を使ってでも計算できるこの単純な方程式が，制御変数 a の値によって全く予測できないような多様な振る舞いを示すことを発見しました．そして，雑誌 Nature にその教育的意義の重要性を感動的に述べています [8]．差分方程式 (3.27) は，一つの制御変数 a を与え， $n = 0$ と置いたとき，適当な初期値 x_0 を与えると x_1 が決まる．次に (3.27) 式で $n = 1$ と置いて今求めた x_1 を右辺に代入すると x_2 が求まる．このような繰り返し計算をすることによって， n が大きくなっ

たとき，時系列 $\{x_n\}$ の特徴を求めることができます． $2.5 \leq a \leq 4.0$ に対する時系列 $\{x_n\}$ の特徴を与えるプログラム例とその出力結果を以下に示します．

```

100 ! 6.1 ロジスティック写像  $x \rightarrow ax(1-x)$ 
110 ! 非線形関数のもつ多様な振る舞い：周期倍分岐
120 ! a: 制御変数
130 DEF f(x)=a*x*(1-x)
140 LET x1=2.5
150 LET x2=4
160 LET y1=-0.2
170 LET y2=1.2
180 SET WINDOW x1, x2, y1, y2
190 ASK PIXEL SIZE(x1, y1; x2, y2) p, q
200 DRAW GRID(0.2, 0.2)
210 SET POINT STYLE 1
220 FOR a=x1 TO x2 STEP (x2-x1)/p
230   LET x=0.3
240   FOR n=1 TO 300
250     LET x=f(x)
260   NEXT n
270   FOR n=301 TO 500
280     LET x=f(x)
290     PLOT POINTS: a, x
300   NEXT n
310 NEXT a
320 END

```

このプログラムでは時系列の初期値 x_0 を 0.3 に選んでいます，分岐図と呼ばれるこの図は，初期値によりません．上記のプログラムにおいて，240-260 は初期値による影響を除去するための操作です．270-300 は同じ a に対して $x_{301} \sim x_{500}$ の 200 個の点をプロットしています． $a < 3.0$ では常に同一点（固定点）をとり， $3.0 < a < 3.45$ では 2 点を交互にとる時系列（2 周期点）に， $3.45 < a < 3.544$ では 4 点を交互にとる 4 周期点，以下同様に，制御変数を大きくしていくに従い，周期が $2^0, 2^1, 2^2, 2^3, \dots$ の




図 3.10: ロジスティック写像の分岐図

ように2倍ずつ増えていく時系列を見いだすことができます。ロジスティック写像は周期倍分岐を与えているのです。そして、周期が 2^∞ を与える集積点の a で時系列はカオスを与えます。垂直な直線上、点で黒く塗りつぶされたところがカオスです。

問題 11 時系列 $\{x_n\}$ の周期倍分岐を確かめるために、いくつかの制御変数 a を与えて、時系列 $\{x_n\}$ の変化を描くプログラムを作成しなさい。

問題 12 ロジスティック写像の分岐図の一部分を拡大して、その図が元の図と類似した構造の図になっていること、拡大図の一部分をさらに拡大すると再び類似した図になっている自己相似性（フラクタルという）を確かめてみよう。

問題 13 写像が正弦関数 $f(x) = a \sin(x)$ で与えられるとき, その分岐図を求めなさい. この結果を見て, 分岐図がロジスティック写像ときわめてよく似た構造をもっていることに気づくだろう.

3.6.2 対数写像

次に全く異なった分岐図を与える非線形写像を紹介してみよう. 写像関数が対数関数

$$x_{n+1} = a + \log(|x_n|) \quad (3.28)$$

で定義される場合 [10] を考えます. ロジスティック写像のプログラム例にならって, その分岐図を求めるプログラムを以下に示します.

```

100 ! 6.2 対数写像 x = a + log|x|
110 ! 非線形関数のもつ多様な振る舞い: 間歇性カオスの出現
120 ! a: 制御変数
130 DEF f(x)=a+LOG(ABS(x))
140 LET x1=-3
150 LET x2=2
160 LET y1=-8
170 LET y2=4
180 SET WINDOW x1, x2, y1, y2
190 ASK PIXEL SIZE(x1, y1; x2, y2) p, q
200 DRAW grid
210 SET POINT STYLE 1
220 FOR a=x1 TO x2 STEP (x2-x1)/p
230   LET x=0.2
240   FOR n=1 TO 5000
250     LET x=f(x)
260   NEXT n
270   FOR n=1 TO 200
280     LET x=f(x)
290     PLOT POINTS: a, x
300   NEXT n
310 NEXT a
320 END

```




図 3.11: 対数写像の分岐図

上記のプログラム 240-260 において 5000 回もの繰り返し計算（ロジスティック写像のときは 300 回）を行っているのは，初期値の影響を消してより鮮明な分岐図を得るための操作です．このプログラムによる分岐図は図 3.11 のように $|a| > 1$ で固定点， $|a| < 1$ においてカオスが出現することを示しています．この分岐図をながめていただけでは，与えられた制御変数 a に対して， $\{x_n\}$ がどのような時系列を表しているのか分かりません．そこで，カオス領域の一方の端に近い (a) $a = -0.95$ と中心付近の (b) $a = 0$ とカオス領域の他方の端に近い (c) $a = 0.95$ の時系列 $\{x_n\}$ を調べてみると図 3.12 のようになります．

問題 14 図 3.12 のような対数写像の時系列を与えるプログラムを作りなさい．




図 3.12: 対数写像の時系列

3.6.3 間歇性カオス

$a = 0$ における時系列の振る舞いは、デタラメでいかにもカオスという感じがしますが、カオス領域の両端ではかなり規則性をもった時系列になっていることが分かります。その状況をもう少し詳しく見ると (a) $a = -0.95$ では小さな振動が単調に増幅してあるところで急に乱れた状態が生じています。規則的な運動部分をラミナー、乱れた状態部分をバーストと呼んでいます。時系列 $\{x_n\}$ はラミナーとバーストが交互に繰り返す構造になっています。ラミナーの長さは長いものから短いものまで様々ですが、実は、 a が更にカオスの端、 -1 に近づくに従いラミナーの平均の長さは、徐々に長くなっていきます。(c) $a = 0.95$ の状況もラミナー部分が (a) の場合の振動とは違って、単調に減少しているだけで、時系列がラミナーとバーストの繰り返しになっていることは共通しています。そして、 a がカオスの端、 1 に近づくに従い、ラミナーが長くなる点も同じです。このようにし

て、固定点領域（規則運動）に近づくに従い、その近傍ではカオスと云ってもかなり組織だった運動をしていることが分かります。この近傍のカオスは間歇性カオス [11] と呼ばれますが、Y. Pomeau と P. Manneville にしたがって (a) の場合のカオスをタイプ 3、(c) の場合のカオスをタイプ 1 と呼んで区別しています。2 つのタイプの間歇性カオスを単純な対数写像で見いだしたのですが、このような規則運動から乱れた運動に変化するとき、間歇性カオスの存在は自然の中に、容易に見いだすことができます。例えば、火山活動や地震、或いは水道の蛇口からしたたり落ちる水滴の運動、化学反応系や熱対流など枚挙にいとまありません。

3.6.4 不規則運動の中の規則性

ロジスティック写像や対数写像、共に単純な数式の規則で時系列 $\{x_n\}$ が作られていくので決定論的ですが、その数式は非線形であるために、先の数列がどのように変化するのか全く予測が付きません。これに対してサイコロ振りには特に細工をしない限り、決定論的な要素はないから同じデタラメでもカオスとは呼びません。ところで、前節でタイプ 1、タイプ 3 の間歇性カオスについて、共に、不規則な長さのラミナーと短いバーストが繰り返し現れる構造をもっていることを学びました。

ラミナーの長さの平均値 l は制御変数 a の関数としてどのような関数で変化していくのか数値解析で求めてみよう。興味があるのは、規則的な運動（定常状態）から不規則運動（カオス状態）に移行する領域の振る舞いですから $|a| = 1$ 近傍を詳しく調べることにします。制御変数を $a = 1 - \varepsilon$ 或いは $a = -1 + \varepsilon$ とおいて、 l の ε による変化を数値解析してみます。プログラミングは少し複雑になります。また、 l も非常に大きくなりますから時系列を $10^6 \sim 10^7$ もとらなければならないことや平均値の精度を上げるために、ラミナーの個数も増やす必要があります。このように計算量が多いので、パソコンの CPU が大きくなったとは云え、かなりの時間が

かかります。以下に $a = -1$ 近傍の平均ラミナー長 l の計算プログラム例を示します。

```

100 ! 6.3 不規則運動の中の規則性：平均ラミナー長の計算
110 DIM J(0 TO 800000) ! J(K): K という長さのラミナーの個数
120 LET MM=50000 ! ラミナーの個数の最大値
130 PRINT " 平均ラミナー長: a=-1 近傍: average=";
140 PRINT MM
150 FOR EP=0.00009 TO 0.00001 STEP -0.00001 ! EP:
160 LET AL=-1+EP ! AL: 制御変数 a
170 LET x=0.2
180 LET AA=0
190 LET S=0.0
200 FOR I=0 TO 600000 ! 終値はラミナー長さの最大値
210 LET J(I)=0
220 NEXT I
230 PRINT
240 PRINT "Epsilon = ";
250 PRINT USING "--%.#####":EP;
260 PRINT " Alpha = ";
270 PRINT USING "--%.#####":AL
280 FOR M=1 TO MM
290 FOR K=1 TO 300000
300 LET y=LOG(x)+AL ! 300-380 ラミナー長の判定
310 LET x=ABS(y)
320 LET yy=LOG(x)+AL
330 LET x=ABS(yy)
340 LET A=ABS(y-yy)
350 LET D=A-AA
360 LET E=ABS(D/A)
370 LET AA=A
380 IF E <= 1.0E-15 OR D >0 THEN
390 GOTO 450
400 ELSE
410 LET KK=2*(K-1)
420 LET J(KK)=J(KK)+1
430 GOTO 460
440 END IF

```

```
450     NEXT K
460     NEXT M
470     FOR K=2 TO 600000 STEP 2
480         IF K>500000 AND J(K)>0 THEN
490             PRINT USING "#####":K;
500             PRINT USING "#####":J(K)
510         END IF
520         LET S=S+K*J(K)
530     NEXT K
540     PRINT "平均ラミナー長=";
550     PRINT USING "-----%.##":S/(MM-J(0))
560     PRINT "Sample(Laminar)の個数=";
570     PRINT USING "#####":MM-J(0)
580 NEXT EP
590 END
```

図 3.13 の結果は、6.3 のプログラム例を用いて得られたデータをもとに、ラミナー長さの ε 依存性を示したものです。平均ラミナー長さ l は $a = -1$ 近傍で $\varepsilon^{-0.496}$ 、 $a = -1$ 近傍で $\varepsilon^{-0.501}$ のべき則に従って ε の減少と共に無限大になることを示しています。タイプ 1 やタイプ 3 の間歇性カオスのラミナー長さが、共に $l \propto \varepsilon^{-0.5}$ に従うことが示されました。このべき則は今、対数写像をモデルに選んで求めましたが、モデルによらないという普遍則をもっています。

問題 15 上記プログラム例で、パラメータ ε を変えて（繰り返し回数やラミナーの個数も変更）実行してみよう。




図 3.13: ラミナー長さはベキ則に従って変化している

関連図書

- [1] 例えば 高等学校「物理 IB」啓林館
- [2] 例えば 高等学校「物理 II」啓林館
- [3] 例えば 「基礎物理学」原 康夫，学術図書出版社
- [4] 「力学 I」原島 鮮，裳華房
- [5] 「計算物理 I」藪下 信，地人書館
- [6] 「天体力学入門」長沢 工，地人書館
- [7] 「カオス入門」長島 弘之，馬場良和，培風館
- [8] Simple mathematical models with very complicated dynamics, R. M. May, Nature, 261(1976)
- [9] 「カオス力学入門」下条 隆嗣，近代科学者
- [10] Intermittent Chaos Generated by Logarithmic Map, T. Kawabe and Y. Kondo, Prog. Theor. Phys. 86,581(1991)
- [11] 「カオスの中の秩序」ペルジユ,P.，ポモウ,Y. 他（相澤 洋二訳），産業図書

